ReflexiveHull_2D

Let R be an arbitrary ring. For a left[1] R-module M the evaluation map $M \xrightarrow{\varepsilon} M^{**}$ is natural, where $^* = \mathrm{Hom}_R(-,R)$ is the dualizing functor. M is called *reflexive* if the evaluation map $M \xrightarrow{\varepsilon} M^{**}$ is an isomorphism. So in general one is intereseted in the kernel and cokernel of ε.

Now let $P_1 \xrightarrow{\varphi} P_0 \to M \to 0$ be an exact sequence with P_1 and P_0 finitely generated projective modules. We call such a sequence a *finite projective presentation* of M. Following [Aus66] we define for M its so-called Auslander dual module $M^\top := \mathrm{coker}(P_0^* \xrightarrow{\varphi^*} P_1^*)$. The isomorphism class of $M^\top$ depends on the presentation of M, but its projective equivalence class depends only on (the isomorphism class of) M (cf. [AB69] and [Eis95, Exercise A3.22]). Further, the so-called evaluation sequence

$$0 \to \mathrm{Ext}^1_R(M^\top, R) \to M \xrightarrow{\varepsilon} M^{**} \to \mathrm{Ext}^2_R(M^\top, R) \to 0$$

is exact (cf. [Aus66] and [HS97, Exercise IV.7.3]). Hence: $\ker(\varepsilon) = \mathrm{Ext}^1_R(M^\top, R)$ and $\mathrm{coker}(\varepsilon) = \mathrm{Ext}^2_R(M^\top, R)$.

In [AB69] a module M with a finite projective presentation is called *k-torsion* free if

$$\mathrm{Ext}^i_R(M^\top, R) = 0 \text{ for } 1 \leq i \leq k.$$

Hence, a module M with a finite projective presentation is reflexive, iff M is 2-torsion free.

For rings where

(Refl) for a finitely generated module M its double-dual M^{**} is reflexive,

one may call M^{**} the *reflexive hull* of M. For commutative noetherian rings this is the case (for a noncommutative two-side noetherian ring see [AB69, the Propositions on p. 143f and the Remark on p. 145]: we need a ring where whenever $\mathrm{Ext}^2_R(N, R)$ is projective for a finitely generated module N, then $\mathrm{Ext}^2_R(N, R) = 0$).

From now on let R be a left Ore (not necessarily commutative) domain (e.g. a left noetherian integral domain). It now makes sense to speak about the torsion submodule $\mathrm{t}(M)$ of the left module M. For a module M with a finite projective presentation there is a natural isomorphism $\mathrm{t}(M) = \mathrm{Ext}^1_R(M^\top, R)$ (cf. [AB69, Pal70, Kas95] and [Qua99] for Ore domains[2]). Hence, torsion freeness is equivalent to 1-torsion freeness. On the other hand, for a finitely generated R-module M, torsion freeness is equivalent to M being a submodule of some free R-module of finite rank.

We consider a torsion free module M, which we can think of as embedded in the free module $M \overset{\iota}{\hookrightarrow} F := R^a$ for some finite a. The diagram

$$\begin{array}{ccc} M & \xrightarrow{\iota} & F \\ {\scriptstyle \varepsilon_M}\downarrow & & \Big\| {\scriptstyle \varepsilon_F} \\ M^{**} & \xrightarrow{\iota^{**}} & F^{**} \end{array}$$

enables one to also embed M^{**} in $F = F^{**}$. In case R additionally satisfies (Refl) then by this one can compute the reflexive hull of M in R^a: $M \leq M^{\mathrm{refl}} = M^{**} \leq R^a$.

```
> restart;
> #with(QuillenSuslin): with(Involutive): with(homalg):
> with(Involutive): with(homalg):
```

We force the Maple ring package `Involutive` to use the external program `JB`, which is a C++ implementation of the involutive division algorithm:

```
> InvolutiveOptions("C++");
```

[1] Unless stated otherwise, we always consider left modules.

[2] The proof in [Qua99] is reproduced in [PQ03, Corollary 2, p. 12], which is available online.

Specify the homalg-table of the ring package Involutive with the QuillenSuslin extension:

```
> #RPQ:=`InvolutiveQS/homalg`;
```

Use the ring package Involutive with the QuillenSuslin extension as the default ring package:

```
> #`homalg/default`:=RPQ;
```

Specify the homalg-table of the ring package Involutive:

```
> RPI:=`Involutive/homalg`;
```

$$RPI := Involutive/homalg$$

Use the ring package Involutive as the default ring package:

```
> `homalg/default`:=RPI;
```

$$homalg/default := Involutive/homalg$$

Define the ring $R = \mathbb{Q}[x, y]$:

```
> var:=[x,y];
```

$$var := [x, y]$$

The rows of the following matrix are the generators of the module M as a submodule of the free module $F := R^{1\times 2}$ of rank 2:

```
> gen := matrix([[-y*(x-y), y*(x^2-x-2*y+2)], [-x*(x-y),
x*(x^2-x-2*y+2)], [x^2*y^2+x^3+x^2*y+x^2+2*x*y-2*y^2,
x*y^3-x*y^2-y^3-2*x^2-x*y+4*y^2-2*y],
[x^3*y-x^3+2*x*y^2-y^3-x^2-4*x*y+4*y^2, 3*y^3+2*x^2-x*y-10*y^2+6*y],
[x^4+2*x^3+2*x^2*y-2*x*y^2-x^2+5*x*y-3*y^2,
3*x*y^2-y^3-4*x^2-6*x*y+6*y^2+4*x-4*y]]);
```

$$gen := \begin{bmatrix} -y\,(x-y) & y\,(x^2-x-2\,y+2) \\ -x\,(x-y) & x\,(x^2-x-2\,y+2) \\ x^2\,y^2+x^3+x^2\,y+x^2+2\,x\,y-2\,y^2 & x\,y^3-x\,y^2-y^3-2\,x^2-x\,y+4\,y^2-2\,y \\ x^3\,y-x^3+2\,x\,y^2-y^3-x^2-4\,x\,y+4\,y^2 & 3\,y^3+2\,x^2-x\,y-10\,y^2+6\,y \\ x^4+2\,x^3+2\,x^2\,y-2\,x\,y^2-x^2+5\,x\,y-3\,y^2 & 3\,x\,y^2-y^3-4\,x^2-6\,x\,y+6\,y^2+4\,x-4\,y \end{bmatrix}$$

The modules $M := \operatorname{coker}\left(R^{1\times 2} \xrightarrow{\mathtt{M}} R^{1\times 5}\right)$, i.e. it is a module on 5 generators with 4 relations:

```
> M:=Image(gen,var);
```

$$\begin{aligned} M := [[[1,\,0,\,0,\,0,\,0] &= [-x\,y+y^2,\, x^2\,y-x\,y-2\,y^2+2\,y], \\ [0,\,1,\,0,\,0,\,0] &= [-x^2+x\,y,\, x^3-x^2-2\,x\,y+2\,x], [0,\,0,\,1,\,0,\,0] = \\ &[x^2\,y^2+x^3+x^2\,y+x^2+2\,x\,y-2\,y^2,\, x\,y^3-x\,y^2-y^3-2\,x^2-x\,y+4\,y^2-2\,y], \\ [0,\,0,\,0,\,1,\,0] &= \\ &[x^3\,y-x^3+2\,x\,y^2-y^3-x^2-4\,x\,y+4\,y^2,\, 3\,y^3+2\,x^2-x\,y-10\,y^2+6\,y], \\ [0,\,0,\,0,\,0,\,1] &= [x^4+2\,x^3+2\,x^2\,y-2\,x\,y^2-x^2+5\,x\,y-3\,y^2, \\ &3\,x\,y^2-y^3-4\,x^2-6\,x\,y+6\,y^2+4\,x-4\,y]], [[x,\,-y,\,0,\,0,\,0], \\ &[y^2-y,\,0,\,-x,\,y-x,\,y], [y,\,-2\,x-3\,y,\,-y,\,x^2+x,\,-x\,y+x+y], \\ &[y,\,y^3-y^2-2\,x-3\,y,\,-x^2-y,\,x\,y+x,\,x+y]], \text{“Presentation”}, \\ &s+1+\frac{s^2}{1-s}+\frac{2\,s}{1-s}+\frac{2}{1-s}+\frac{2}{(1-s)^2},\, [11,\,2]] \end{aligned}$$

Compute the embedding of $M \overset{\iota}{\hookrightarrow} F$ (this might differ from the above matrix of generators, since the procedure Image tries to find fewer generators of the image[3].):

```
> iota:=ImageEmb(gen,var);
```

[3] Unless invoked with the option "USE_IMAGE_OF_GENERATORS"

$$\iota := \begin{bmatrix} -x\,y + y^2 & x^2\,y - x\,y - 2\,y^2 + 2\,y \\ -x^2 + x\,y & x^3 - x^2 - 2\,x\,y + 2\,x \\ x^2\,y^2 + x^3 + x^2\,y + x^2 + 2\,x\,y - 2\,y^2 & x\,y^3 - x\,y^2 - y^3 - 2\,x^2 - x\,y + 4\,y^2 - 2\,y \\ x^3\,y - x^3 + 2\,x\,y^2 - y^3 - x^2 - 4\,x\,y + 4\,y^2 & 3\,y^3 + 2\,x^2 - x\,y - 10\,y^2 + 6\,y \\ x^4 + 2\,x^3 + 2\,x^2\,y - 2\,x\,y^2 - x^2 + 5\,x\,y - 3\,y^2 & 3\,x\,y^2 - y^3 - 4\,x^2 - 6\,x\,y + 6\,y^2 + 4\,x - 4\,y \end{bmatrix}$$

The free module F (the target of ι):

```
> F:=FreeHullModule(iota,var);
```

$$F := [[[1,\,0] = [1,\,0],\,[0,\,1] = [0,\,1]],\,[[0,\,0]],\,\text{“Presentation”},\,\frac{2}{(1-s)^2},\,[0,\,2]]$$

ι is indeed an embedding:

```
> IsHom(M,iota,F,var);
```

$$\mathit{true}$$

```
> IsInjective(M,iota,F,var);
```

$$\mathit{true}$$

The module M is torsion free (since it was constructed as a submodule of a free module):

```
> IsTorsionFree(M,var);
```

$$\mathit{true}$$

But not reflexive:

```
> IsReflexive(M,var);
```

$$\mathit{false}$$

Compute the double-dual M^{**} (although we don't need this computation in what follows). We know that M^{**} is reflexive and hence free, since the (right) global dimension of R variables is 2. But the nice thing is that the following computation yields M^{**} on two free generators. `homalg` won't always be able to find the free generators, and in this case one might want to use the ring package `Involutive` in conjunction with the package `QuillenSuslin` as indicated above:

```
> HHM:=HomHom_R(M,var);
```

$$\mathit{HHM} := [[[1,\,0] = \begin{bmatrix} 0 \\ 1 \end{bmatrix},\,[0,\,1] = \begin{bmatrix} 1 \\ 0 \end{bmatrix}],\,[[0,\,0]],\,\text{“Presentation”},\,\frac{2}{(1-s)^2},\,[0,\,2]]$$

M^{**} is reflexive (as expected):

```
> IsReflexive(HHM,var);
```

$$\mathit{true}$$

But M^{**} is stably free (even free):

```
> IsStablyFreeFFR(HHM,var);
```

$$\mathit{true}$$

The image of M in F again given by a finite presentation:

```
> Image(iota,F,var,"USE_IMAGE_OF_GENERATORS");
```

$$\begin{aligned}
&[[[1,\,0,\,0,\,0,\,0] = [-x\,y + y^2,\,x^2\,y - x\,y - 2\,y^2 + 2\,y],\\
&[0,\,1,\,0,\,0,\,0] = [-x^2 + x\,y,\,x^3 - x^2 - 2\,x\,y + 2\,x], [0,\,0,\,1,\,0,\,0] =\\
&[x^2\,y^2 + x^3 + x^2\,y + x^2 + 2\,x\,y - 2\,y^2,\,x\,y^3 - x\,y^2 - y^3 - 2\,x^2 - x\,y + 4\,y^2 - 2\,y],\\
&[0,\,0,\,0,\,1,\,0] =\\
&[x^3\,y - x^3 + 2\,x\,y^2 - y^3 - x^2 - 4\,x\,y + 4\,y^2,\,3\,y^3 + 2\,x^2 - x\,y - 10\,y^2 + 6\,y],\\
&[0,\,0,\,0,\,0,\,1] = [x^4 + 2\,x^3 + 2\,x^2\,y - 2\,x\,y^2 - x^2 + 5\,x\,y - 3\,y^2,\\
&3\,x\,y^2 - y^3 - 4\,x^2 - 6\,x\,y + 6\,y^2 + 4\,x - 4\,y]], [[x,\,-y,\,0,\,0,\,0],\\
&[y^2 - y,\,0,\,-x,\,y - x,\,y],\,[y,\,-2\,x - 3\,y,\,-y,\,x^2 + x,\,-x\,y + x + y],\\
&[y,\,y^3 - y^2 - 2\,x - 3\,y,\,-x^2 - y,\,x\,y + x,\,x + y]],\,\text{“Presentation”},\\
&s + 1 + \frac{s^2}{1-s} + \frac{2\,s}{1-s} + \frac{2}{1-s} + \frac{2}{(1-s)^2},\,[11,\,2]]
\end{aligned}$$

The induced map $M^{**} \xrightarrow{\mu:=\iota^{**}} F^{**} = F$ (compare μ, which is the embedding of M^{**} into F, with ι, which is the embedding of M into F):

```
> mu:=HomHomMap_R(M,iota,F,var);
```

$$\mu := \begin{bmatrix} x+1 & y-2 \\ 2\,x-y+1 & 3\,y-4-x^2+x \end{bmatrix}$$

Compute the image of M^{**} inside F, i.e. the reflexive hull of $M \leq F$ in F (the image of M^{**} in F is again given on free generators):

```
> HH:=Image(mu,F,var,"USE_IMAGE_OF_GENERATORS");
```

$$HH := [[[1,\,0] = [x+1,\,y-2],\,[0,\,1] = [2\,x-y+1,\,3\,y-4-x^2+x]],\,[[0,\,0]],$$
$$\text{“Presentation”},\,\frac{2}{(1-s)^2},\,[0,\,2]]$$

```
> IsReflexive(HH,var);
```

$$true$$

Author: Mohamed Barakat

Date:	2007-07-24
Last modified:	2007-07-30 11:40

References

[AB69] Maurice Auslander and Mark Bridger, *Stable module theory*, Memoirs of the American Mathematical Society, No. 94, American Mathematical Society, Providence, R.I., 1969. MR MR0269685 (42 #4580) 1

[Aus66] Maurice Auslander, *Coherent functors*, Proc. Conf. Categorical Algebra (La Jolla, Calif., 1965), Springer, New York, 1966, pp. 189–231. MR MR0212070 (35 #2945) 1

[BCG+03] Y. A. Blinkov, C. F. Cid, V. P. Gerdt, W. Plesken, and D. Robertz, *The* MAPLE *Package* Janet*: I. Polynomial Systems.* II. *Linear Partial Differential Equations*, Proc. 6th Int. Workshop on Computer Algebra in Scientific Computing, Passau, Germany, 2003, (`http://wwwb.math.rwth-aachen.de/Janet`).

[BR] Mohamed Barakat and Daniel Robertz, `homalg` – *A meta-package for homological algebra*, submitted. `arXiv:math.AC/0701146` and (`http://wwwb.math.rwth-aachen.de/homalg`).

[BR07] ———, `homalg` *project*, 2004-2007, (`http://wwwb.math.rwth-aachen.de/homalg`).

[Eis95] David Eisenbud, *Commutative algebra*, Graduate Texts in Mathematics, vol. 150, Springer-Verlag, New York, 1995, With a view toward algebraic geometry. MR MR1322960 (97a:13001) 1

[Fab07] Anna Fabianska, `QuillenSuslin`*: A* Maple *package to compute a free basis of a projective module over the polynomial ring*, (`http://wwwb.math.rwth-aachen.de/QuillenSuslin`), 2006-2007.

[FQ07] Anna Fabianska and Alban Quadrat, *Applications of the Quillen-Suslin theorem in multidimensional systems theory*, Proceedings of the Workshop on Gröbner bases in Control Theory and Signal Processing, Radon Book Series, de Gruyter, 2007, To appear.

[HS97] P. J. Hilton and U. Stammbach, *A course in homological algebra*, second ed., Graduate Texts in Mathematics, vol. 4, Springer-Verlag, New York, 1997. MR MR1438546 (97k:18001) 1

[Kas95] Masaki Kashiwara, *Algebraic study of systems of partial differential equations*, Mém. Soc. Math. France (N.S.) (1995), no. 63, xiv+72. MR MR1384226 (97f:32012) 1

[Pal70] V. P. Palamodov, *Linear differential operators with constant coefficients*, Translated from the Russian by A. A. Brown. Die Grundlehren der mathematischen Wissenschaften, Band 168, Springer-Verlag, New York, 1970. MR MR0264197 (41 #8793) 1

[PQ03] Jean-François Pommaret and Alban Quadrat, *A functorial approach to the behaviour of multidimensional control systems*, Int. J. Appl. Math. Comput. Sci. **13** (2003), no. 1, 7–13, (`http://www-sop.inria.fr/cafe/Alban.Quadrat/Publications/amcs.pdf`) Multidimensional systems nD and iterative learning control (Czocha Castle, 2000),. MR MR1986488 (2004d:93026) 1

[Qua99] Alban Quadrat, *Analyse algébrique des systèmes de contrôle linéaires multidimensionnels*, Ph.D. thesis, Ecole Nationale des Ponts et Chaussées, CERMICS, France, September 1999. 1

Lehrstuhl B für Mathematik, RWTH-Aachen University, 52062 Germany
E-mail address: mohamed.barakat@rwth-aachen.de